

**Example 1c** — Testing coefficients across class models

[Description](#)[Remarks and examples](#)[Also see](#)

## Description

In this example, we demonstrate how to use `test` and `contrast` to test the equality of coefficients across classes after fitting an FMM.

## Remarks and examples

stata.com

We continue with [Example 1b](#), where we fit a three-component mixture model for the logarithm of medical expenditures. The best model we found was one in which we used total chronic conditions (`totchr`) in the `lcprob()` option of `fmm` to predict latent class probabilities and additional covariates to predict the means for the latent classes.

At this point, we may want to begin looking at how the effect of covariates differs by class. For example, we may want to know if being female has the same effect on medical expenditures in the low-, medium-, and high-spending classes. To do this, we can test the coefficient on `1.sex` in the equations for the class means.

Many of Stata's postestimation commands require you to specify an expression if you want, for example, to perform a test of equality (`test`), compute a difference between estimates (`lincom`), or compute a ratio of coefficients (`nlcom`). Remembering how to specify the names of estimates can be difficult. We first redisplay the estimation output with the `coeflegend` option so we can see the legend of the coefficients and how to specify them in an expression.

```
. fmm, coeflegend
Finite mixture model                               Number of obs = 2,955
Log likelihood = -4712.3871
```

|         | Coefficient | Legend             |
|---------|-------------|--------------------|
| 1.Class |             | (base outcome)     |
| 2.Class |             |                    |
| totchr  | .9376084    | _b[2.Class:totchr] |
| _cons   | -.6114399   | _b[2.Class:_cons]  |
| 3.Class |             |                    |
| totchr  | 1.16097     | _b[3.Class:totchr] |
| _cons   | -3.270603   | _b[3.Class:_cons]  |

## 2 Example 1c — Testing coefficients across class models

```
Class: 1
Response: lmedexp
Model: regress
```

|                | Coefficient | Legend                          |
|----------------|-------------|---------------------------------|
| lmedexp        |             |                                 |
| income         | .0048917    | _b[lmedexp:1.Class#c.income]    |
| age            | .0261976    | _b[lmedexp:1.Class#c.age]       |
| c.age#c.age    | -.0000843   | _b[lmedexp:1.Class#c.age#c.age] |
| totchr         | .5412491    | _b[lmedexp:1.Class#c.totchr]    |
| sex            |             |                                 |
| Female         | .1793964    | _b[lmedexp:1.sex#1.Class]       |
| _cons          | 5.035174    | _b[lmedexp:1.Class]             |
| var(e.lmedexp) | 2.311098    | _b[/var(e.lmedexp)#1.Class]     |

(output omitted)

Here we test individually whether the effect of being female in class 1 is the same as the effect of being female in class 2 and whether the effect of being female in class 2 is the same as the effect of being female in class 3.

```
. test (_b[lmedexp:1.Class#1.sex] = _b[lmedexp:2.Class#1.sex])
( 1) [lmedexp]1.sex#1bn.Class - [lmedexp]1.sex#2.Class = 0
      chi2( 1) = 3.04
      Prob > chi2 = 0.0811
. test (_b[lmedexp:2.Class#1.sex] = _b[lmedexp:3.Class#1.sex])
( 1) [lmedexp]1.sex#2.Class - [lmedexp]1.sex#3.Class = 0
      chi2( 1) = 1.46
      Prob > chi2 = 0.2270
```

Neither test is significant; therefore, we cannot reject the null of the coefficients being equal. We can also do a joint test.

```
. test (_b[lmedexp:1.Class#1.sex] = _b[lmedexp:2.Class#1.sex])
> (_b[lmedexp:2.Class#1.sex] = _b[lmedexp:3.Class#1.sex])
( 1) [lmedexp]1.sex#1bn.Class - [lmedexp]1.sex#2.Class = 0
( 2) [lmedexp]1.sex#2.Class - [lmedexp]1.sex#3.Class = 0
      chi2( 2) = 5.11
      Prob > chi2 = 0.0775
```

The joint test is also not significant.

Alternatively, `contrast` can do all the work for us without the need of remembering coefficient names. Here we use the `a.` operator on `Class` to compare adjacent class categories. See [\[R\] contrast](#) for additional comparisons that we could make.

```
. contrast sex#a.Class, equation(lmedexp)
Contrasts of marginal linear predictions
Margins: asbalanced
```

|                  | df | chi2 | P>chi2 |
|------------------|----|------|--------|
| lmedexp          |    |      |        |
| sex#Class        |    |      |        |
| (joint) (1 vs 2) | 1  | 3.04 | 0.0811 |
| (joint) (2 vs 3) | 1  | 1.46 | 0.2270 |
| Joint            | 2  | 5.11 | 0.0775 |

We obtain exactly the same results reported by `test` but in a more succinct format.

## Also see

- [FMM] [fmm intro](#) — Introduction to finite mixture models
- [FMM] [fmm: regress](#) — Finite mixtures of linear regression models
- [FMM] [fmm postestimation](#) — Postestimation tools for fmm